
Latest versions of Spike2 and Signal 

Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.05  Signal version 3.10 

 Spike2 version 5.19  Signal version 2.16 

 Spike2 demo  Signal demo 

Spike2   

 

Q. I am recording three channels of EMG activity from the jaw and would like to 
calculate the mean of the rectified EMG activity from this data. 

A. You can create arithmetic combinations of channels in Spike2 using the Virtual 
channel functions from the Analysis menu. Virtual channels are based on 
expressions which can be a combination of channel values (in the form of Ch(n) 
where n is a channel number) and standard arithmetic functions. To create a 
channel holding the mean of the rectified EMG activity in this example we can 
use the expression (Abs(Ch(1))+Abs(Ch(2))+Abs(Ch(3)))/3, where the 
Abs() function returns the absolute value (rectification) of each channel. We can 
match the virtual channel sample interval to one of our existing channels from the 
Match to channel drop-down list in the dialog. 

You can also apply rectification using the Channel Process commands from the Analysis menu



 

 
Virtual channel dialog and mean of rectified channel data displayed in channel v1 

You can generate your own equations or re-use previously entered expressions 
by clicking on the double arrow button to the right of the expression field and 
selecting options from the dialog below. For further information see the virtual 
channels section in the on-line help. 

 
Expression options 

  

  

 

Q. I am interested in your 3304 constant current stimulator and would like to know if 
the settings could be controlled directly from Signal. 

A. The CED 3304 is an isolated current stimulator that can be directly controlled by 
Signal as an auxiliary states device. If you select support for the 3304 during 
installation of Signal you can access a configuration dialog for the stimulator from 
within the States tab of the sampling configuration. The current magnitude range 
setting needs to be set to match the rotary switch position on the front panel of 
the 3304 unit. You can then control the current intensity from the set-up dialog 
and assign different settings to multiple state outputs for use during your 

Support options for auxiliary devices such as the 3304, MagStims and the MultiClamp 700 amplifier can be selected during the Signal installation



experiment. These stimulus parameters are automatically saved in the sampled 
frame as variables that can be accessed by the script language. 

 

 
CED 3304 constant current stimulator 

 
CED 3304 stimulator configuration 

 

 

 

Scripts: Spike2  

Q. Is there a script available that would allow me to create a new data file from two 
separate files, either by copying channels or amalgamating data from one to the 
end of another?    

A. The attached script, MergeFiles.s2s, can be used to merge files by copying 
channels from a source file to a destination file using a shared timebase or 
append one file to the end of another. The user is prompted for the files to open 
when the script is run and can select whether to merge or append from a dialog 
in the script.   

 

 Scripts: Signal  

Q. I have a series of evoked response data files recorded over a long trial period 
and saved in a common directory. What I would like to do is create auto-
averages for each file based on n frames per average and then amalgamate all 
of these auto-average results in a single file. It would also be important to know 
which frames in the amalgamated average file related to which data file. Is there 
any way to achieve this? 

A. The attached script, BatchAutoAverage.sgs, is a batch processing example 
script that creates auto-averages from every data file in a nominated directory. It 
creates a new memory view that amalgamates all of the frames from the auto-
averages from each data file. Each frame in the new memory view has an 
associated frame comment holding the name of the source data file and which 
source frames were used to create the average. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

scripts in this newsletter 

'MergeFiles.s2s|Script to append files together or copy channels from one file to another

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 29/11/07

DoToolbar();

Func DoToolbar();
ToolbarClear();
ToolbarSet(1,"&Quit||Quit the script");
ToolbarSet(3,"Merge files||Copy channels from one file and save them to new channels in a destination file",MergeFiles%);
ToolbarSet(4,"Append files||Copy channels and append them to the same channel no.s in the destination file",AppendFiles%);
return Toolbar("Script to merge or append data files",1023);
end;

Func MergeFiles%()
var sourceFile%;			            'Handle for Source file
var destFile%;			                'Handle for Destination file
var freeChans%[401];	                'Array to hold number of free channels available in the Destination file
var sourceChans%[401];	                'Array to hold number of Source channels
var allChans%[401];

destFile% := FileOpen("",0,1,"Open Destination file");  'Prompt to open destination file
ChanList(freeChans%[], 128);	        'Get number of free channels in dest file

sourceFile% := FileOpen("",0,1,"Open file to copy"); 'Prompt to open source file
ChanList(sourceChans%[]);		        'Get total number of channels in source file

if destFile% or sourceFile% = 0 then
    Message("Error opening files!");
    halt;
endif;

if sourceChans%[0] > freeChans%[0] then	'Check that enough channels are available in the file
    Message("Not enough available channels in the destination file!\nExport the destination file as a new data file and increase\nthe number of channels!");
    halt;
endif;

View(sourceFile%);
ChanSave(-1, 0, destFile%);	            'Save all channels to destination file
View(destFile%);
ChanList(allChans%[]);
ChanShow(allChans%[]);
FrontView(DestFile%);			        'Bring destination file to the front
Window(0,0,100,100);			        'Maximise the window.
return 1;
end;

Func AppendFiles%()
var tsep := 1;                                      'separation time between data files
var ok%;
var fh%,dh%;                                        'file handles of the source and destination files
var tim;

fh% := FileOpen("*.smr", 0, 0,"Open Destination file");                     'open file1
dh% := FileNew(7, 1, View(fh%).Binsize()*1e6, 1, 200);  'make output 

View(fh%);                                          'copy File1
tim := MaxTime();
ChanSave(-1, -1, dh%);                              'copy it all
FileClose(0,-1);
   
fh% := FileOpen("*.smr", 0, 0, "Open file to append");                      'open file2 and append
View(fh%);
ChanSave(-1, -1, dh%, 0, MaxTime(), tim+tsep);
FileClose(0,-1);

View(dh%);ChanShow(-1);Draw(0, MaxTime());          'Display the result
return 1;
end;









CED
MergeFiles.s2s


'$BatchAutoAverage|Example to open all data files in a directory and create an auto average
'Each frame of the auto-averages from all files are then amalgamated into a single memory view
'Frame comments are used to tell the user the source file and frames for each frame in the new file

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 29/11/07

var files$[20];		'Declare array to hold up to 20 files
var chans%[32];		'Array to hold channels in files
var numFiles%;			'Number of files found
var c%,d%;				'counters to loop through files and channels
var numChans%;			'Number of channels in current file
var success%;			'Test if file opened correctly
var ok%;
var fPerAve%;
var fh%;
var flag%;
var result%;
var num%:=1;

FilePathSet("",0,"Select directory of files for analysis");		'Prompts user to select directory
numFiles%:= FileList(files$[],0);		'Fill array with up to 20 files and return number of files found
if numFiles%<1 then							'If no files found...
	Message("No data files in directory!");		'Display message and end script
halt;endif;

DlgCreate("Multi frame average");  			'Start new dialog
DlgInteger(1, "Frames per average", 2, 10000);
ok% := DlgShow(fPerAve%);    				'ok is 0 if user cancels
if ok% < 1 then
	Message("Script cancelled");
Halt;endif;

for c%:=1 to numFiles% do			'Loop through each file
	success%:=FileOpen(files$[c%-1],0);		'Check it opened correctly

	if success% > 0 then			'If it opened correctly..
		ChanList(chans%[]);		'Get a list of the channels in the file
		Window(0,0,100,100);
		WindowVisible(1);			'Maximise the file and make it visible
		XRange(0,Maxtime());		'Display whole time range

		for d%:=1 to chans%[0] do	'Loop through each channel
			ChanShow(d%);				'and make sure it is visible
		next;
		
		fh% := View();                  			'get the starting view handle
		MultiAve(fh%, fPerAve%);	'Function that does the work
		
		Optimise(-1);				'Optimise the data
		'Message("Average created successfully");		'display message
		View(fh%);
		FileClose(0,-1);			'When message acknowledged, close file to keep screen tidy
	else;
		Message("Could not open file %s",files$[c%]);	'If file not opened, display error message
	endif;
	
next;

View(result%).Window(0,0,100,100);		'Display result file

Func MultiAve(f%, fpa%)
var ave%;
var noFr%;
var c%;
var fRange;
var fName$;
var com$;
var fNo$;

View(f%);
fName$:= FileName$(3);		'Get name of source file
fNo$:= FileName$(4);
fRange:= MaxTime();			'Get time range of source file

ave%:=SetAutoAv(2,fPerAve%,fPerAve%,fRange,0.000000,0,0,0,0);	'Create auto-average
ProcessFrames(1,-1,-1,0,1,1);		'process
'WindowVisible(1);
'
noFr%:= FrameCount();		'Get count of frames in auto-average

for c%:=1 to noFr% do

	if flag%=1 then
		View(ave%).Frame(c%);			'Get current frame
		View(result%).AppendFrame();	'Append new frame to result file
		View(result%).Frame(num%);		'make sure it is current
		ArrConst(View(result%,1).[],View(ave%,1).[]);		'Copy data to result file

		if c%=1 then
			com$:= Print$(": Frames %d-%d",c%,fPerAve%);	'get the list of source frames
		else
			com$:= Print$(": Frames %d-%d",((fPerAve%*(c%-1))+1),(fPerAve%*c%));
		endif;

		View(result%).FrameComment$(fName$+fNo$+com$);		'Write to frame comment
		num%+=1;	'Keep track of number of frames in result file
	else
		View(ave%).Frame(c%);		'If this is the first frame
		result%:= SetCopy(-1,"Multi Average",1);	'Create a new memory view for the result
		flag%:=1;	'flag to say we have a result file open
		View(result%).WindowVisible(1);				
		com$:= Print$(": Frames %d-%d",c%,fPerAve%);	'Get source frame info
		View(result%).FrameComment$(fName$+fNo$+com$);	'Write to frame comment
		num%+=1
	endif;
next;

View(ave%);
FileClose(0,-1);	'Close the last auto-average
return 1;
end;

BatchAutoAverage.sgs



 

Did you know…? 
You can overdraw channels by dragging and dragging channel numbers on top of each 
other in a Spike2 and Signal file. These groups of overdrawn channels can then share a 
common Y-axis with an optional offset using the Lock axes checkbox in the Y Range 
dialog. 

 

 

 

X and Y force measurement channels overlaid and sharing a common Y axis 

Recent questions   
Q. Am I able to run Spike2 on an Intel Mac running Parallels virtualization software? 

A. Both Spike2 and Signal will run on an Intel Mac which has Windows XP installed 
using Apple’s own BootCamp software (see eNewsletter 27) or Parallels desktop 
for Mac software by SWsoft. Using BootCamp, Windows XP is installed as a 
separate operating system that can be selected at start-up to run in place of Mac 
OS X. The Parallels desktop software runs both OS X and Windows XP in 
separate “virtual machines”, effectively running both operating systems at once. 
More information on these applications can be found on the Apple and SWsoft 
websites respectively. 

 Both Spike2 v6 and Signal v4 along with a Power1401 mk II and Micro1401 mkII 
have been successfully tested for acquisition and analysis using a Mac Mini 
1.66GHz Intel Core Duo processor with 512MB of memory. 

User group      
If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
 

All Trademarks are acknowledged to be the Trademarks of the registered holders. 

